

Formulación y Evaluación de Proyectos Turísticos – Clase 4



Fabrizio Marcillo Morla MBA

barcillo@gmail.com

(593-9) 4194239



Fabrizio Marcillo Morla

- Guayaquil, 1966.
- BSc. Acuicultura. (ESPOL 1991).
 - Magister en Administración de Empresas. (ESPOL, 1996).
- Profesor ESPOL desde el 2001.
- 20 años experiencia profesional:
 - ◆ Producción.
 - ◆ Administración.
 - ◆ Finanzas.
 - ◆ Investigación.
 - ◆ Consultorías.

[Otras Publicaciones del mismo autor en Repositorio ESPOL](#)

Uso Financiero de Excel


- Entender Uso de Formulas y Funciones Financieras Básicas de Excel:
 - ◆ VAN: Función y Formulas.
 - ◆ TIR: Función y Goal Seek.
 - ◆ Payback Simple y Descontado: Formulas.
 - ◆ Relación Beneficio / Costo: Formulas.
 - ◆ Tabla Amortización: Funciones y Fórmulas.
- Uso práctico de conceptos aprendidos.
 - ◆ Ejemplos: [taller1.xls](#).

fx



Ejercicio

- Se piensa instalar un proyecto en el cual se invertirán \$2,500, y se esperan ingresos netos de \$1,000 anuales por 5 años.
- El costo de oportunidad es de \$20.
- Calcule:
 - ◆ Flujos de caja descontados.
 - ◆ VAN.
 - ◆ TIR.
 - ◆ Periodo de recuperación Simple y Descontado.
 - ◆ Relación beneficio / costo.



Función o Fórmula?

fx

- **Formula:** Ecuación que ejecuta calculos. Pueden referirse a otras celdas o a valores y usan operadores (=,+,-,*,/,^... etc), son totalmente flexibles:
 - ◆ Ventaja: Flexibilidad.
 - ◆ =A4/(1+\$B\$)^A3
- **Función:** Fórmulas especiales predefinidas de Excel que ejecutan cálculos específicos. Funcionan de forma mas o menos fija y establecida:
 - ◆ Ventaja: Facilidad de uso.
 - ◆ =SUM(A4:D4)
 - ◆ =SUMA(A4:D4)
- Se puede combinar funciones y formulas.

VAN: Función

fx

- $=NPV(\text{Tasa}, \text{Rango } VF_1 - VF_n) + VF_0$
- $=VAN(\text{Tasa}, \text{Rango } VF_1 - VF_n) + VF_0$ (o VNA)
- Cuidado con Rango VF_0
- Ejemplo: $r = 20\%$ y Flujo:

	B	C	D	E	F	G	H
1							
2	Año	0	1	2	3	4	5
3	Ingresos		1,000	1,000	1,000	1,000	1,000
4	Egresos	(2,500)					
5	Flujo Neto	(2,500)	1,000	1,000	1,000	1,000	1,000

- $=NPV(20\%, D5:H5) + C5$
- $=VAN(20\%, D5:H5) + C5$

VAN: Función

fx

- Cuidado con rango a utilizar:

	A	B	C	D	E	F	G
1	Año	0	1	2	3	4	5
2	Ingresos		1,000	1,000	1,000	1,000	1,000
3	Egresos	(2,500)					
4	Flujo Neto	(2,500)	1,000	1,000	1,000	1,000	1,000
5	Descontado	(2,500)	833	694	579	482	402
6	Suma	491	(VAN)				
7	r =	20%					
8	VAN =	409	Mal	=NPV(B7,B4:G4)			
9	VAN =	491	OK	=NPV(B7,C4:G4)+B4			

$$VAN = \sum_{n=0}^t \frac{C_n}{(1+r)^n}$$

VAN: Formula

- El VAN se lo puede calcular facilmente usando la formula del VAN:
 - ◆ Crear una formula para descontar cada uno de los flujos de caja futuros:
 - ◆ =VF/(1+\$r\$)^n
 - ◆ (reemplazar VF y n por celda respectiva).
 - ◆ Sumar los flujos de caja descontados:

	A	B	C	D	E	F	G	H
1		Año	0	1	2	3	4	5
2		Flujo Neto	(2,500)	1,000	1,000	1,000	1,000	1,000
3		Tasa de Descuento	20%					
4		Flujo Descontado	(2,500)	833	694	579	482	402
5			=C2/(1+\$C\$3)^C1	=D2/(1+\$C\$3)^D1	=E2/(1+\$C\$3)^E1	=F2/(1+\$C\$3)^F1	=G2/(1+\$C\$3)^G1	=H2/(1+\$C\$3)^H1
6	VAN =	491	=SUM(C4:H4)					

TIR: Función

fx

- =IRR(RangFlujo) o =IRR(RangFlujo,?)
- =TIR(RangFlujo) o =TIR(RangFlujo,?)

	A	B	C	D	E	F	G
1	Año	0	1	2	3	4	5
2	Flujo Neto	(2,500)	1,000	1,000	1,000	1,000	1,000
3	TIR =	29%	=IRR(C2:H2)				

TIR: Función

fx

- ? Es valor inicial para empezar iteración. Normalmente no necesario (asume 10%), si da error #NUM!, cambiar el valor de ? por uno cercano al TIR esperado.
- Excel hace 20 iteraciones y si no logra un resultado exacto a 0.00001% da error #NUM!.

	A	B	C	D	E	F
1		Año	0	1	2	3
2			-300	10	10	10
3		TIR =	#NUM!	=IRR(C2:F2)		
4		TIR =	-63%	=IRR(C2:F2,-0.9)		

$$0 = \sum_{n=0}^t \frac{C_n}{(1+r)^n}$$

TIR: Goal Seek (Buscar Objetivo)

- TIR puede ser calculada usando Herramienta Goal Seek (Buscar Objetivo de Excel):
 - ◆ Hacer Flujo de Caja y Flujo Descontado direccionado a una celda(r) o poner formula de VAN direccionado a una celda(r).
 - ◆ Usar Goal Seek para buscar valor de esta celda r que haga el VAN = 0.

	A	B	C	D	E	F	G
1	Año	0	1	2	3	4	5
2	Flujo Neto	(2,500)	1,000	1,000	1,000	1,000	1,000
3	Descontado	(2,500)	777	604	470	365	284
4	Suma=	0					
5	r =	29%					
6	VAN =	0					
7							
8							
9							
10							
11							
12							

Goal Seek ? X

Set cell:

To value:

By changing cell:

Periodo de Recuperación

- Calcular el Flujo (o Flujo Descontado).
- Calcular el flujo acumulado (flujo acumulado anterior + flujo actual).
- Determinar cuando el flujo se hace positivo (a dedazo o con funciones).
- Se puede usar funciones IF() para determinar cuando cambia el flujo de signos y calcular el número de años y meses del PayBack. Unido a funciones de texto se puede automatizar totalmente el proceso.

Periodo de Recuperación

- Ejemplo de calculo de Periodo de Recuperación totalmente automatizado, usando funciones y formulas de texto.

A	B	C	D	E	F	G	H	I
1	Año	0	1	2	3	4	5	
2	Ingresos		1,000	1,000	1,000	1,000	1,000	
3	Egresos	(2,500)						
4	Flujo Neto	(2,500)	1,000	1,000	1,000	1,000	1,000	
5	Flujo Acumulado	(2,500)	(1,500)	(500)	500	1,500	2,500	
6	2 =IF(ROUND(SUM(C7:H7),0)=12,SUM(C6:H6)+1,SUM(C6:H6))	x	x	2	x	x	x	=IF(ISNUMBER(C7),C1,"x")
7	6 =IF(ROUND(SUM(C7:H7),0)=12,0,SUM(C7:H7))	x	x	6	x	x	x	=IF(AND(C5<0,D5>=0),-C5/((D5-C5)/12),"x")
8	P. R. Simple	2 Años y 6 meses						=IF(H5<0,99,"x")
9		=IF(A7<99,TEXT(A6,0)&" Años y "&TEXT(A7,0)&" meses","No Recupera")						
10	Tasa de Descuento	20%						
11	Flujo Descontado	(2,500)	833	694	579	482	402	
12	Flujo Acumulado Descontado	(2,500)	(1,667)	(972)	(394)	89	491	
13	3 =IF(ROUND(SUM(C14:H14),0)=12,SUM(C13:H13)+1,SUM(C13:H13))	x	x	x	3	x	x	=IF(ISNUMBER(C14),C1,"x")
14	10 =IF(ROUND(SUM(C14:H14),0)=12,0,SUM(C14:H14))	x	x	x	10	x	x	=IF(AND(C12<0,D12>=0),-C12/((D12-C12)/12),"x")
15	P. R. Descontado	3 Años y 10 meses						=IF(H12<0,99,"x")
16		=IF(A14<99,TEXT(A13,0)&" Años y "&TEXT(A14,0)&" meses","No Recupera")						

Relación Beneficio Costo

- Se calcula flujo descontado de Ingresos y Egresos.
- Se calcula el valor actual de los ingresos y egresos (suma).
- Se divide ingresos / -egresos:

	A	B	C	D	E	F	G	H
1								
2	Año	0	1	2	3	4	5	
3	Ingresos		1,000	1,000	1,000	1,000	1,000	
4	Egresos	(2,500)						
5	Flujo Neto	(2,500)	1,000	1,000	1,000	1,000	1,000	
6								
7	Tasa de Descuento		20%					
8								
9	Flujo Descontado							
10	Ingresos	-	833	694	579	482	402	=B3/(1+\$C\$7)^B\$2
11	Egresos	(2,500)	-	-	-	-	-	
12	Flujo Neto	(2,500)	833	694	579	482	402	
13								
14	Total Ingresos VA		2,991	=SUM(B10:G10)				
15	Total Egresos VA	(2,500)	=SUM(B11:G11)					
16	Total Flujo Neto VA		491	(VAN)				
17								
18	Relacion Beneficio Costo		1.20	=-C14/C15				

Amortización Prestamos

fx

- Valor total a pagar:
 - ◆ =PMT(i,Plazo,Monto)
 - ◆ =PAGO(I, Plazo,Monto)
- Valor a pagar intereses:
 - ◆ =IPMT(i,Periodo,Plazo,Monto)
 - ◆ =PAGOI(i,Periodo,Plazo,Monto)
 - ◆ =Remanente Capital * i.
- Valor a Pagar Capital:
 - ◆ =PPMT(i,Periodo,Plazo,Monto)
 - ◆ =PAGOP(i,Periodo,Plazo,Monto)
 - ◆ =Valor Total – Pago Intereses.
- Remanente Capital:
 - ◆ Remanente Anterior – Pago Capital.

Tabla Amortización

- Prestamo de \$1,000,000 a 15% interes y 5 años plazo:

	A	B	C	D	E	F	G	H	I
2	Monto	(1,000,000)							
3	Interes	15%							
4	Plazo	5							
5									
6									
7	Tabla 1								
8	Años	1	2	3	4	5	Total		
9	Pago Total	298,316	298,316	298,316	298,316	298,316	1,491,578	=PMT(\$B\$3,\$B\$4,\$B\$2)	
10	Pago Interes	150,000	127,753	102,168	72,746	38,911	491,578	=IPMT(\$B\$3,B8,\$B\$4,\$B\$2)	
11	Pago Capital	148,316	170,563	196,147	225,569	259,405	1,000,000	=PPMT(\$B\$3,B8,\$B\$4,\$B\$2)	
12									
13									
14	Tabla 2								
15	Años	1	2	3	4	5	Total		
16	Pago Total	298,316	298,316	298,316	298,316	298,316	1,491,578	=PMT(\$B\$3,\$B\$4,\$B\$2)	
17	Pago Interes	150,000	127,753	102,168	72,746	38,911	491,578	=B19*\$B\$3	
18	Pago Capital	148,316	170,563	196,147	225,569	259,405	1,000,000	=B16-B17	
19	Remanente	1,000,000	851,684	681,122	484,974	259,405	(0)	=-B2	=B19-B18